Soluzioni del compito di Calcolo Numerico del 10 ottobre 2003

Enrico Bertolazzi

```
> restart:
> with(LinearAlgebra):
```

Esercizio 1

matrice identita e vettori canonici della base usati per i conti successivi

```
> e1,e2,e3,e4 := <1,0,0,0>,<0,1,0,0>,<0,0,1,0>,<0,0,0,1>;
```

$$e1, e2, e3, e4 := \begin{bmatrix}1\\0\\0\\0\end{bmatrix}, \begin{bmatrix}0\\1\\0\\0\end{bmatrix}, \begin{bmatrix}0\\0\\1\\0\end{bmatrix}, \begin{bmatrix}0\\0\\0\\1\end{bmatrix}$$

Matrice identità

```
> ID := <e1|e2|e3|e4>;
```

$$ID := \begin{bmatrix}1 & 0 & 0 & 0\\0 & 1 & 0 & 0\\0 & 0 & 1 & 0\\0 & 0 & 0 & 1\end{bmatrix}$$

Matrice iniziale

```
> A := Matrix([[-3, -7, -4, -2],
               [-4, -10, -2, -4],
               [-6, 6, -6, -6],
               [12, 12, 12, 12]]);
```

$$A := \begin{bmatrix}-3 & -7 & -4 & -2\\-4 & -10 & -2 & -4\\-6 & 6 & -6 & -6\\12 & 12 & 12 & 12\end{bmatrix}$$

```
> A0 := A :
```

Matrice di permutazione

```
> P1 := <e4|e2|e3|e1>;
```

$$P1 := \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

Scambio la prima con la quarta riga

```
> P1A0 := P1.A0;
```

$$P1A0 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ -4 & -10 & -2 & -4 \\ -6 & 6 & -6 & -6 \\ -3 & -7 & -4 & -2 \end{bmatrix}$$

Matrice di eliminazione

```
> L1 := ID-<0,P1A0[2,1],P1A0[3,1],P1A0[4,1]>.Transpose(e1)/P1A0[1,1];
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{1}{3} & 1 & 0 & 0 \\ \frac{1}{2} & 0 & 1 & 0 \\ \frac{1}{4} & 0 & 0 & 1 \end{bmatrix}$$

Primo passo del metodo di Gauss

```
> A1 := L1.P1A0;
```

$$A1 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & -6 & 2 & 0 \\ 0 & 12 & 0 & 0 \\ 0 & -4 & -1 & 1 \end{bmatrix}$$

Matrice di scambio seconda e terza riga

```
> P2 := <e1|e3|e2|e4>;
```

$$P2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Scambio la seconda con la terza riga

```
> P2A1 := P2.A1;
```

$$P2A1 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & 12 & 0 & 0 \\ 0 & -6 & 2 & 0 \\ 0 & -4 & -1 & 1 \end{bmatrix}$$

Matrice di eliminazione

```
> L2 := ID-<0,0,P2A1[3,2],P2A1[4,2]>.Transpose(e2)/P2A1[2,2];
```

$$L2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & \frac{1}{2} & 1 & 0 \\ 0 & \frac{1}{3} & 0 & 1 \end{bmatrix}$$

Secondo Passo del metodo di Gauss

```
> A2 := L2.P2A1;
```

$$A2 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & 12 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & -1 & 1 \end{bmatrix}$$

Nessuno scambio

```
> P3 := <e1|e2|e3|e4>;
```

$$P3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Nessuno scambio

```
> P3A2 := P3.A2;
```

$$P3A2 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & 12 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & -1 & 1 \end{bmatrix}$$

Matrice di eliminazione

```
> L3 := ID-<0,0,0,P3A2[4,3]>.Transpose(e3)/P3A2[3,3];
```

$$L3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & \frac{1}{2} & 1 \end{bmatrix}$$

Terzo passo del metodo di Gauss

```
> A3 := L3.P3A2;
```

$$A3 := \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & 12 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Matrici P, L, U

```
> P := P3.P2.P1:
  L := P.(L3.P3.L2.P2.L1.P1)^(-1):
  U := A3:
  P, L, U ;
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{-1}{2} & 1 & 0 & 0 \\ \frac{-1}{3} & \frac{-1}{2} & 1 & 0 \\ \frac{-1}{4} & \frac{-1}{3} & \frac{-1}{2} & 1 \end{bmatrix}, \begin{bmatrix} 12 & 12 & 12 & 12 \\ 0 & 12 & 0 & 0 \\ 0 & 0 & 2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Controllo

```
> R := P.A - L.U;
```

$$R := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Calcolo la soluzione del primo problema

```
> b  := < -25, -34, -12, 72>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} -25 \\ -34 \\ -12 \\ 72 \end{bmatrix}, \begin{bmatrix} 72 \\ -12 \\ -34 \\ -25 \end{bmatrix}, \begin{bmatrix} 72 \\ 24 \\ 2 \\ 2 \end{bmatrix}, \begin{bmatrix} 1 \\ 2 \\ 1 \\ 2 \end{bmatrix}$$

Calcolo la soluzione del secondo problema

```
> b  := < 3, 8, -12, 0>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} 3 \\ 8 \\ -12 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -12 \\ 8 \\ 3 \end{bmatrix}, \begin{bmatrix} 0 \\ -12 \\ 2 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ -1 \\ 1 \\ 0 \end{bmatrix}$$

Esercizio 2

Punti di interpolazione:

```
> x := 'x' ;
```

$$x := x$$

```
> X := [-2,-1,0,2,3];
  Y := [15,0,-1,15,80];
```

$$X := [-2, -1, 0, 2, 3]$$

$$Y := [15, 0, -1, 15, 80]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$z^4 - 1$$

Costruzione delle differenze divise di ordine 0

```
> f1 := Y[1];
  f2 := Y[2];
  f3 := Y[3];
  f4 := Y[4];
  f5 := Y[5];
```

$$f1 := 15$$
$$f2 := 0$$
$$f3 := -1$$
$$f4 := 15$$
$$f5 := 80$$

Differenze divise

```
> f12 := (f2-f1)/(X[2]-X[1]);
  f23 := (f3-f2)/(X[3]-X[2]);
  f34 := (f4-f3)/(X[4]-X[3]);
  f45 := (f5-f4)/(X[5]-X[4]);
```

$$f12 := -15$$
$$f23 := -1$$
$$f34 := 8$$
$$f45 := 65$$

Differenze divise seconde

```
> f123 := (f23-f12)/(X[3]-X[1]);
  f234 := (f34-f23)/(X[4]-X[2]);
  f345 := (f45-f34)/(X[5]-X[3]);
```

$$f123 := 7$$
$$f234 := 3$$
$$f345 := 19$$

Differenze divise terze

```
> f1234 := (f234-f123)/(X[4]-X[1]);
  f2345 := (f345-f234)/(X[5]-X[2]);
```

$$f1234 := -1$$
$$f2345 := 4$$

Differenze divise quarte

```
> f12345 := (f2345-f1234)/(X[5]-X[1]);
```

$$f12345 := 1$$

Polinomi della base

```
> w0 := 1 ;
  w1 := x-X[1] ;
  w2 := expand(w1 * ( x - X[2])) ;
  w3 := expand(w2 * ( x - X[3])) ;
  w4 := expand(w3 * ( x - X[4])) ;
```

$$w0 := 1$$

$$w1 := x + 2$$

$$w2 := x^2 + 3x + 2$$

$$w3 := x^3 + 3x^2 + 2x$$

$$w4 := x^4 + x^3 - 4x^2 - 4x$$

Polinomio interpolante

```
> p := f1*w0 + f12 * w1 + f123 * w2 + f1234 * w3 + f12345 * w4 ;
```

$$p := -1 + x^4$$

Esercizio 4

Funzione da integrare:

```
> f := x -> (1+x^2)*cos(x/2) ;
```

$$f := x \rightarrow (1 + x^2)\cos\left(\frac{1}{2}x\right)$$

Integrale esatto

```
> a,b := 0,2:
  int(f(x),x=a..b);
  esatto := evalf(%);
```

$$-6\sin(1) + 16\cos(1)$$

$$esatto := 3.596010985$$

Derivata quarta della funzione

```
> ddddf := (D@@4)(f) ;
  collect(ddddf(x),cos(x/2)) ;
```

$$ddddf := x \rightarrow -3\cos\left(\frac{1}{2}x\right) + x\sin\left(\frac{1}{2}x\right) + \frac{1}{16}(1 + x^2)\cos\left(\frac{1}{2}x\right)$$

$$\left(-\frac{47}{16} + \frac{1}{16}x^2\right)\cos\left(\frac{1}{2}x\right) + x\sin\left(\frac{1}{2}x\right)$$

Possibile costante che maggiore il modulo della derivata seconda

```
> M := 47/16 + 4/16 + 2 ;
```

$$M := \frac{83}{16}$$

Verifica grafica:

```
> plot(abs(ddddf(x)),x=a..b) ;
```



Calcolo in numero minimo stimato di intervalli

```
> err := 10^(-5) = (b-a)^5/(180*n^4)*M ;
```

$$err := \frac{1}{100000} = \frac{83}{90\ n^4}$$

```
> isolate(err,n^4) ;
```

$$n^4 = \frac{830000}{9}$$

```
> evalf(sqrt(sqrt(rhs(%))));
```

$$17.42644884$$

Intervalli stimati: 18

```
> n := 18 ;
```

$$n := 18$$

```
> h := (b-a)/n ;
```

$$h := \frac{1}{9}$$

```
> simp := evalf((h/3)*(f(a)+f(b)+
                     4*sum(f(a+h*(2*i-1)),i=1..(n/2))+
                     2*sum(f(a+h*2*i),i=1..(n/2-1))));
```

$$simp := 3.596007917$$

```
> simp - esatto ;
```

$$-0.000003068$$

Calcolo con 6 intervalli

```
> h := (b-a)/6 ;
```

$$h := \frac{1}{3}$$

```
> simp := evalf((h/3)*(f(a)+f(b)+
                      4*sum(f(a+h*(2*i-1)),i=1..3)+
                      2*sum(f(a+h*2*i),i=1..2)));
```

$$simp := 3.595760325$$

```
> simp - esatto ;
```

$$-0.000250660$$

Esercizio 4

Matrice del sistema

```
> L  := <<0,0,-1>|<0,0,-1>|<0,0,0>>;
  DG := <<4,0,0>|<0,2,0>|<0,0,2>>;
  U  := <<0,0,0>|<-1,0,0>|<-1,-1,0>>;
  A  := DG+L+U ;
```

$$L := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ -1 & -1 & 0 \end{bmatrix}$$

$$DG := \begin{bmatrix} 4 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

$$U := \begin{bmatrix} 0 & -1 & -1 \\ 0 & 0 & -1 \\ 0 & 0 & 0 \end{bmatrix}$$

$$A := \begin{bmatrix} 4 & -1 & -1 \\ 0 & 2 & -1 \\ -1 & -1 & 2 \end{bmatrix}$$

Termine noto

```
> b := <1,-1,1>;
```

$$b := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix}$$

libera x, y, z

```
> x := 'x' ;
  y := 'y' ;
  z := 'z' ;
```

$$x := x$$

$$y := y$$

$$z := z$$

Metodo Iterativo

```
> iter := (DG+L).<x[k+1],y[k+1],z[k+1]> + U.<x[k],y[k],z[k]> - b ;
```

$$iter := \begin{bmatrix} 4\, x_{k+1} - y_k - z_k - 1 \\ 2\, y_{k+1} - z_k + 1 \\ -x_{k+1} - y_{k+1} + 2\, z_{k+1} - 1 \end{bmatrix}$$

```
> isolate(iter[1]=0,x[k+1]);
  isolate(iter[2]=0,y[k+1]);
  isolate(iter[3]=0,z[k+1]);
```

$$x_{k+1} = \frac{1}{4} y_k + \frac{1}{4} z_k + \frac{1}{4}$$

$$y_{k+1} = \frac{1}{2} z_k - \frac{1}{2}$$

$$z_{k+1} = \frac{1}{2} x_{k+1} + \frac{1}{2} y_{k+1} + \frac{1}{2}$$

Matrice di iterazione -(D+L)^(-1)U

```
> MI := -(DG+L)^(-1).U;
```

$$MI := \begin{bmatrix} 0 & \frac{1}{4} & \frac{1}{4} \\ 0 & 0 & \frac{1}{2} \\ 0 & \frac{1}{8} & \frac{3}{8} \end{bmatrix}$$

Polinomio caratteristico della matrice di iterazione

```
> chpoly := CharacteristicPolynomial(MI,z);
```

$$chpoly := z^3 - \frac{3}{8}z^2 - \frac{1}{16}z$$

```
> res := solve(chpoly,z);
```

$$res := 0, \frac{1}{2}, \frac{-1}{8}$$

Calcolo il raggio spettrale

```
> seq(abs(res[i]),i=1..3);
```

$$0, \frac{1}{2}, \frac{1}{8}$$

Raggio spettrale = 1/2 < 1 ==> il metodo converge.

Faccio alcune iterate:

```
> P0 := <2,1,2> ;
```

$$P0 := \begin{bmatrix} 2 \\ 1 \\ 2 \end{bmatrix}$$

```
> P1 := (DG+L)^(-1).(b-U.P0):
  P1, evalf(P1) ;
```

$$\begin{bmatrix} 1 \\ \frac{1}{2} \\ \frac{5}{4} \end{bmatrix}, \begin{bmatrix} 1. \\ 0.5000000000 \\ 1.250000000 \end{bmatrix}$$

```
> P2 := (DG+L)^(-1).(b-U.P1) :
  P2, evalf(P2) ;
```

$$\begin{bmatrix} \frac{11}{16} \\ \frac{1}{8} \\ \frac{29}{32} \end{bmatrix}, \begin{bmatrix} 0.6875000000 \\ 0.1250000000 \\ 0.9062500000 \end{bmatrix}$$

Calcolo i residui

```
> R1 := b - A . P1 :
  R2 := b - A . P2 :
  R1, evalf(R1) ;
  R2, evalf(R2) ;
```

$$\begin{bmatrix} \frac{-5}{4} \\ \frac{-3}{4} \\ 0 \end{bmatrix}, \begin{bmatrix} -1.250000000 \\ -.7500000000 \\ 0. \end{bmatrix}$$

$$\begin{bmatrix} \frac{-23}{32} \\ \frac{-11}{32} \\ 0 \end{bmatrix}, \begin{bmatrix} -.7187500000 \\ -.3437500000 \\ 0. \end{bmatrix}$$

Norma infinito

```
> max(seq(abs(R1[i]),i=1..3));
  max(seq(abs(R2[i]),i=1..3));
  evalf(%%),evalf(%) ;
```

$$\frac{5}{4}$$

$$\frac{23}{32}$$

$$1.250000000, 0.7187500000$$

Norma 1

```
> add(abs(R1[i]),i=1..3);
  add(abs(R2[i]),i=1..3);
  evalf(%%),evalf(%) ;
```

$$2$$

$$\frac{17}{16}$$

$$2., 1.062500000$$

Norma 2

```
> sqrt(add(abs(R1[i])^2,i=1..3));
  sqrt(add(abs(R2[i])^2,i=1..3));
  evalf(%%),evalf(%) ;
```

$$\frac{1}{4}\sqrt{34}$$

$$\frac{5}{32}\sqrt{26}$$

$$1.457737974, 0.7967217991$$

```
>
```